NONMEM Implementation of Cell Life Span Models for Hematological Drug Effects

Juan Jose Perez Ruixo,

Hui C. Kimko, Andrew Chow, Vladmir Piotrovsky

Johnson&Johnson
PHARMACEUTICAL RESEARCH
& DEVELOPMENT, L.L.C.

Outline

- Cell life span concept
- NM implementation
  - Colony stimulating drug effect
  - Anticancer drug effect
- NM performance of a life span model
- Summary

Cell Life Span Model

- **Several PKPD models have been used to model drug effects on natural cells, but only a few implemented the cell life span concept.**

  - Krzyzanski W, et al. Basic Pharmacodynamic Model for Agents that Alter Production of Natural Cells. *JPP 1*999; 27: 311-337.

  - Krzyzanski W, et al. Model of Hematologic Effects of Anticancer Agents. *JPP* 2002: 29: 311-337.

Life Span Model & Drug Effects



- **Cells survive for a specific duration of time (cell life span) and then disappear by conversion to another type of cells.**
- **Cell life span models assume that the rate of cell loss is equal to the production rate but is delayed by the cell lifespan.**

NM Implementation of Colony Stimulating Drug Effects

Stimulatory Drug Effect: Zero-order input model (k_0)



$$\frac{dR}{dt} = k_0 \cdot S(t) - k_0 \cdot S(t - T_R)$$

$$S(t) = 1 + \frac{S_{max} C(t)^{\gamma}}{SC_{50}{}^{\gamma} + C(t)^{\gamma}}$$

$$S(t - T_R) = 1 + \frac{S_{max} C(t - T_R)^{\gamma}}{SC_{50}{}^{\gamma} + C(t - T_R)^{\gamma}}$$

$$Baseline: \; R_0 = k_0 \cdot T_R$$



Life Span Model (k_0):

Delay compartment



$$\frac{dR}{dt} = k_0 \cdot \left[1 + \frac{S_{max} C(t)^{\gamma}}{SC_{50}{}^{\gamma} + C(t)^{\gamma}}\right] - k_0 \cdot \left[1 + \frac{S_{max} C(t - T_R)^{\gamma}}{SC_{50}{}^{\gamma} + C(t - T_R)^{\gamma}}\right]$$



Life Span Model (k_0): NM code

NM code for IDR model:

```
$MODEL
  COMP = CENTRAL
  COMP = (EFFECT,DEFOBS)


$PK
 CL    =  THETA(1) * EXP(ETA(1))
 V     =  THETA(2) * EXP(ETA(2))
 KEL   =  CL/V
 SMAX  =  THETA(3) * EXP(ETA(3))
 SC50  =  THETA(4) * EXP(ETA(4))
 KIN   =  THETA(5) * EXP(ETA(5))
 KOUT  =  THETA(6) * EXP(ETA(6))
 F2    =  KIN/KOUT

$DES
 DADT(1) = -KEL*A(1)


 EFF1 = 1 + SMAX*A(1)/(SC50+A(1))


 DADT(2) = KIN*EFF1 - KOUT*A(2)
```

NM code for LS model (K_0) :

```
$MODEL
  COMP = CENTRAL
  COMP = (EFFECT,DEFOBS)
  COMP = DCENTRAL

$PK CALLFL = -2
 CL    =  THETA(1) * EXP(ETA(1))
 V     =  THETA(2) * EXP(ETA(2))
 KEL   =  CL/V
 SMAX  =  THETA(3) * EXP(ETA(3))
 SC50  =  THETA(4) * EXP(ETA(4))
 K0    =  THETA(5) * EXP(ETA(5))
 ALAG3 =  THETA(6) * EXP(ETA(6))
 F2    =  K0*ALAG3   ; R0 = K0 *TR

$DES
 DADT(1) = -KEL*A(1)
 DADT(3) = -KEL*A(3)

 ST1 = 1 + SMAX*A(1)/(SC50+A(1))
 ST3 = 1 + SMAX*A(3)/(SC50+A(3))

 DADT(2) = K0 * ST1 - K0 * ST3
```

S(t) S(t-T_R)

Life span model (k_0): NM code

#ID	TIME	AMT	DV	CMT
1	0	10	0	1
1	0	1	0	2
1	**0**	**10**	**0**	**3**
1	**0.25**	**0**	**0.1**	**2**
1	**0.5**	**0**	**1.2**	**2**

CALLFL = -2: Call the subroutine with every event record and with additional and lagged doses

NM code for LS model (K_0) :

```
$MODEL
  COMP = CENTRAL
  COMP = (EFFECT,DEFOBS)
  COMP = DCENTRAL

$PK CALLFL = -2
 CL    =  THETA(1) * EXP(ETA(1))
 V     =  THETA(2) * EXP(ETA(2))
 KEL   =  CL/V
 SMAX  =  THETA(3) * EXP(ETA(3))
 SC50  =  THETA(4) * EXP(ETA(4))
 K0    =  THETA(5) * EXP(ETA(5))
 ALAG3 =  THETA(6) * EXP(ETA(6))
 F2    =  K0*ALAG3   ; R0 = K0 *TR

$DES
 DADT(1) = -KEL*A(1)
 DADT(3) = -KEL*A(3)

 ST1 = 1 + SMAX*A(1)/(SC50+A(1))
 ST3 = 1 + SMAX*A(3)/(SC50+A(3))

 DADT(2) = K0 * ST1 - K0 * ST3
```

S(t) S(t-T_R)

Stimulatory Drug Effect: First-order Input Model (k_1)



$$\frac{dP}{dt} = k_0 \cdot S(t) - k_1 \cdot P(t)$$

$$\frac{dR}{dt} = k_1 \cdot P(t) - k_1 \cdot P(t - T_R)$$

Baseline:

$$P_0 = \frac{k_0}{k_1}$$

$$R_0 = K_0 \cdot T_R$$



Life Span Model (k_1)



$$\frac{dP}{dt} = k_0 \cdot S(t) - k_1 \cdot P(t)$$

$$\frac{dR}{dt} = k_1 \cdot P(t) - k_1 \cdot P(t - T_R)$$



Life span model (k_1): NM code

```
$MODEL
  COMP = CENTRAL; drug conc.
  COMP = PREC   ; precursor
  COMP = (RESP,DEFOBS) ;3
  COMP = DCENTRAL      ;4
  COMP = DPREC         ;5

$PK CALLFL = -2
;*** PK/PD PARAMETERS ***
 CL    =  THETA(1) * EXP(ETA(1))
 V     =  THETA(2) * EXP(ETA(2))
 KEL   =  CL/V
 SMAX  =  THETA(3) * EXP(ETA(3))
 SC50  =  THETA(4) * EXP(ETA(4))

;*** SYSTEM PARAMETERS ***
 K0    =  THETA(5) * EXP(ETA(5))
 K1    =  THETA(6) * EXP(ETA(6))
 ALAG4 =  THETA(7) * EXP(ETA(7))
 ALAG5 =  ALAG4

;*** INITIAL CONDITIONS ***
 F2    =  K0/K1
 F3    =  K0 * ALAG3
 F5    =  F2
```





Life span model (k_1):

NM code continue





#ID	TIME	AMT	DV	CMT
1	0	10	0	1
1	0	1	0	2
1	**0**	**1**	**0**	**3**
1	**0**	**10**	**0**	**4**
1	**0**	**1**	**0**	**5**
1	0.25	0	1	3

```
$DES

;*** CENTRAL + PREC COMP (IDR)***
 DADT(1) = -KEL*A(1)
 ST1     = 1 + SMAX*A(1)/(SC50+A(1))
 DADT(2) = K0 * ST1 - K1 * A(2)

;*** DCENTRAL + DPREC COMP ***
 DADT(4) = -KEL*A(4)
 ST4     = 1 + SMAX*A(4)/(SC50+A(4))
 DADT(5) = K0 * ST4 - K1 * A(5)

;if t < ALAG5
 A5 = F5   ; = F2 =  K0/K1
 IF (T.GT.ALAG5) A5 = A(5)

;*** R COMP ***
 DADT(3) = K1 * A(2) - K1 * A5
```

P(t) P(t - T_R)

NM Implementation of Anticancer Drug Effects

Anticancer Drug Effect

Multiple-Pool Cell Lifespan Model

Krzyzanski W, et al. Model of Hematologic Effects of Anticancer Agents. *JPP* 2002: 29: 311-337.



Fig. 1. Model for hematologic effects of anticancer agents. The progenitor cells are produced at the zero-order rate k_{in} and they reside in the mitotic pool P for time T_P. In this compartment the anticancer agent kills the cell with the function $f(C)$ where C is the drug plasma concentration. The surviving cells enter the maturation pool M where they stay for time T_M. The mature cells are released to the circulation pool R from which they are lost after time T_R.

Linear Cytotoxicity: ***f(C) = k·C***

Nonlinear Cytotoxicity: $$f(C) = \frac{K_{max} C}{K C_{50} + C}$$

Anticancer Drug Effect: Model

- **EQUATIONS**

$e^0 = 1$: no killing effect, all converted to M

$e^{\infty} = 0$: full killing effect, none converted to M

fraction, probability

$$\frac{dP}{dt} = k_{in} - k_{in} \cdot exp\left(- \int_{t-T_P}^{t} f(C(z))dz \right) - f(C(t)) \cdot P$$

$$\frac{dM}{dt} = k_{in} \cdot exp\left(- \int_{t-T_P}^{t} f(C(z))dz \right) - k_{in} \cdot exp\left(- \int_{t-T_P-T_M}^{t-T_M} f(C(z))dz \right)$$

$$\frac{dR}{dt} = k_{in} \cdot exp\left(- \int_{t-T_P-T_M}^{t-T_M} f(C(z))dz \right) - k_{in} \cdot exp\left(- \int_{t-T_P-T_M-T_R}^{t-T_M-T_R} f(C(z))dz \right)$$

- **The challenge is to calculate** $\int_{t-T_P}^{t} f\left[C(z)\right] \cdot dz$



- $\int_{t-T_P-T_M}^{t-T_M} f(C(z))dz$ **and** $\int_{t-T_P-T_M-T_R}^{t-T_M-T_R} f(C(z))dz$ **are** $\int_{t-T_P}^{t} f\left[C(z)\right] \cdot dz$ **delayed by TM and TM+TR**

Anticancer Drug Effect: Integrals (I)

$$\int_{t-T_P}^{t} f[C(z)] \cdot dz = \int_{0}^{t} f[C(z)] \cdot dz - \int_{0}^{t-T_P} f[C(z)] \cdot dz$$

e.g., $T_P = 5$

If $t \leq T_P$, $\int_{0}^{t-T_P} f[C(z)] \cdot dz = 0$ ⟶ $\int_{t-T_P}^{t} f[C(z)] \cdot dz = \int_{0}^{t} f[C(z)] \cdot dz$





Anticancer Drug Effect: Integrals (II)

If $t > T_P$,

$$\int_{t-T_P}^{t} f[C(z)] \cdot dz = \int_{0}^{t} f[C(z)] \cdot dz - \int_{0}^{t-T_P} f[C(z)] \cdot dz$$



e.g., $T_P = 5$

$\therefore$ **AUC(t) – AUC (before t-Tp)**

Anticancer Drug Effect: NM Code

```
$MODEL
  COMP = CENTRAL
  COMP = AUC
  COMP = DCENTRAL
  COMP = DAUC
  COMP = (RESP,DEFOBS)

$PK CALLFL = -2
;*** PK PARAMETERS ***
 CL    =  THETA(1) * EXP(ETA(1))
 V     =  THETA(2) * EXP(ETA(2))
 KEL   =  CL/V

;*** PD PARAMETERS ***
 KMAX  =  THETA(3) * EXP(ETA(3))
 SC50  =  THETA(4) * EXP(ETA(4))

;*** SYSTEM PARAMETERS ***
 K0    =  THETA(5) * EXP(ETA(5))
 ALAG3 =  THETA(6) * EXP(ETA(6))
 ALAG4 =  ALAG3
;*** INITIAL CONDITIONS ***
 F5    =  K0 * ALAG3
```

```
$DES

;*** CENTRAL + AUC COMP ***
 DADT(1) = - KEL*A(1)
 FC1     = KMAX*A(1)/(SC50+A(1))
 DADT(2) = FC1

;*** DCENTRAL + DAUC COMP ***
 DADT(3) = - KEL*A(3)
 FC2     = KMAX*A(3)/(SC50+A(3))
 DADT(4) = FC2

;*** AUC - DAUC ***
 DIFF = A(2) - A(4)
 PR   = EXP(-DIFF)

;*** P COMP ***
 DADT(5) = K0 - K0*PR - FC1*A(5)
```

$$\frac{dP}{dt} = k_{in} - k_{in} \cdot exp\left(- \int_{t-T_P}^{t} f(C(z))dz \right) - f(C(t)) \cdot P$$



Anticancer Drug Effect: NM Code

```
$MODEL
  COMP = CENTRAL
  COMP = AUC
  COMP = DCENTRAL
  COMP = DAUC
  COMP = (RESP,DEFOBS)

$PK CALLFL = -2
;*** PK PARAMETERS ***
 CL    =  THETA(1) * EXP(ETA(1))
 V     =  THETA(2) * EXP(ETA(2))
 KEL   =  CL/V

;*** PD PARAMETERS ***
 KMAX  =  THETA(3) * EXP(ETA(3))
 SC50  =  THETA(4) * EXP(ETA(4))

;*** SYSTEM PARAMETERS ***
 K0    =  THETA(5) * EXP(ETA(5))
 ALAG3 =  THETA(6) * EXP(ETA(6))
 ALAG4 =  ALAG3
;*** INITIAL CONDITIONS ***
 F5    =  K0 * ALAG3
```

```
$DES

;*** CENTRAL + AUC COMP ***
 DADT(1) = - KEL*A(1)
 FC1     = KMAX*A(1)/(SC50+A(1))
 DADT(2) = FC1

;*** DCENTRAL + DAUC COMP ***
 DADT(3) = - KEL*A(3)
 FC2     = KMAX*A(3)/(SC50+A(3))
 DADT(4) = FC2

;*** AUC - DAUC ***
 DIFF = A(2) - A(4)
 PR   = EXP(-DIFF)

;*** P COMP ***
 DADT(5) = K0 - K0*PR - FC1*A(5)
```

#ID	TIME	AMT	DV	CMT
1	0	10	0	1
1	0	0	0	2
1	**0**	**10**	**0**	**3**
1	**0**	**0**	**0**	**4**
1	**0**	**1**	**0**	**5**
1	0.25	0	1	5



NM Performance Evaluation

NM Performance Evaluation

- **DATA**
  - Full PD profile available for 25 subjects per dose level
  - Four dose levels: 1, 10, 100 and 1000 mg
- **METHOD: Simulation + Estimation using WINGS for NM**
- **MODEL**
  - PK: One-compartment IV model
  - PD:



$$\frac{dR}{dt} = k_0 \cdot S(t) - k_0 \cdot S(t - T_R)$$

$$S(t) = 1 + \frac{S_{max} C(t)^{\gamma}}{SC_{50}{}^{\gamma} + C(t)^{\gamma}}$$

- **PARAMETERS:**
  - PK parameters: CL and V fixed to arbitrary values.
  - PD parameters: SMAX, SC50 w/o IIV. $\gamma = 1$.
  - System parameters: KIN, TR w IIV

NM Performance: FOCE + TBS



NM Performance: FO + TBS



NM Performance: FO















Summary

- Mixed-effects models implementing cell life span concept can be used to deal with the dynamics of natural cells when colony stimulating drugs and anticancer drugs are administered.

- Cell life span models combined with other PKPD models add more flexibility to model the drug response.

- The performance of NM with cell life span models is similar to that observed with other models: FOCE and TBS should be used when feasible.

Thank you!

Complete NM codes are available electronically upon request.
[Email: jperezru@prdbe.jnj.com]

Life Span vs. Transit CMT Models

- The cell life span concept can be described by the empirical transit compartment model

- However, the cell life span model is more physiology based; and leads to faster NM convergence
  - When data of different cell types are available (RET & RBC, or MK & PL), cell life span models are preferable

- NM allows maximum 20 lagged doses. ADDL and II features should be used to avoid the limitation